<?xml version="1.0" encoding="Windows-1251"?>
<xsl:stylesheet version="1.0" xmlns:xsl="http://www.w3.org/1999/XSL/Transform">

    <xsl:output method="html" version="1.0" encoding="Windows-1251" media-type="text/xhtml"/>

    <xsl:strip-space elements="*"/>

    <xsl:template match="text()|@*">
        <xsl:value-of select="."/>
    </xsl:template>

    <xsl:template match="comment()"/>

    <xsl:template match="/">
        <html><body><xsl:apply-templates select="//uri"/></body></html>
    </xsl:template>

    <xsl:template match="uri">
        <p><strong>
        <xsl:apply-templates/>
        </strong></p>
    </xsl:template>

    <xsl:template match="uri//protocol">
        <xsl:apply-templates select="text()"/>
        <xsl:text>://</xsl:text>
    </xsl:template>

    <xsl:template match="uri//auth">
        <xsl:apply-templates/>
        <xsl:text>@</xsl:text>
    </xsl:template>

    <xsl:template match="uri//pass|uri//port">
        <xsl:text>:</xsl:text>
        <xsl:apply-templates select="text()"/>
    </xsl:template>

    <xsl:template match="uri//path">
        <xsl:text>/</xsl:text>
        <xsl:apply-templates/>
    </xsl:template>

    <xsl:template match="uri//dir">
        <xsl:apply-templates/>
        <xsl:text>/</xsl:text>
    </xsl:template>

    <xsl:template match="uri//script/type">
        <xsl:text>.</xsl:text>
        <xsl:apply-templates/>
    </xsl:template>

    <xsl:template match="uri//query">
        <xsl:text>?</xsl:text>
        <xsl:apply-templates/>
    </xsl:template>

    <xsl:template match="uri//param[position()>1]">
        <xsl:text>&amp;</xsl:text>
        <xsl:apply-templates/>
    </xsl:template>

    <xsl:template match="uri//param/value">
        <xsl:text>=</xsl:text>
        <xsl:apply-templates/>
    </xsl:template>

    <xsl:template match="uri//fragment">
        <xsl:text>#</xsl:text>
        <xsl:apply-templates/>
    </xsl:template>

</xsl:stylesheet>